

第2部分：《太阳能光热发电原理、技术及数值方法》的有关源程序

程序 1 槽式聚光器光学性能模拟的 MCRT 程序及 MCRT-FVM 耦合处理

1.1 物理模型

槽式聚光器是广泛应用于光热发电的太阳能聚光器，该聚光器中的太阳辐射传输过程涵盖了光热发电系统所涉及的大多数典型光学过程。鉴于此，可将槽式聚光器光学性能模拟作为学习蒙特卡罗光线追迹 (Monte Carlo ray tracing, MCRT) 方法的入门算例。

下面以一种简化的槽式聚光器为例，来介绍 MCRT 在模拟聚光器光学性能时的应用。该槽式聚光器由抛物面槽式反射镜和吸热管组成，其结构示意图参见图 1。反射镜的几何聚光比为 20、宽度为 4398.23 mm、长度为 4000 mm、边缘角为 90° 、焦距为 1099.57 mm、反射率为 1、形面误差为 0，吸热管外半径为 35 mm、长度为 4000 mm、吸收率为 1。假设太阳视半径为 7.5 mrad，入射阳光的主光轴垂直于反射镜进光口平面，阳光直射辐照度为 $1000 \text{ W}\cdot\text{m}^{-2}$ 。




图 1 简化的槽式光学模型及所用直角坐标系示意图

为了大家学习方便，这里给出了用于模拟槽式聚光器光学性能的程序 TOPS(Trough Optical Performance Simulation)的源代码，参见 1.2。

在模拟中，TOPS 需要用到两个直角坐标系。第一个坐标系是如图 1 所示的 XYZ

系，其原点(O)位于吸热管一端的中心处， X 与 Y 轴分别沿反射镜的长度和宽度方向， Z 轴指向天顶；另一坐标系是以入射光线与镜面的交点(P)为原点的局部坐标系 $X_1Y_1Z_1$ ，其中 X_1 轴与 X 轴同向， Y_1 轴沿理想抛物面型线的切线方向， Z_1 轴沿理想抛物面型线的法线方向，参见图 1 所示。TOPS 的源代码采用 Fortran 90 编写，支持的编译器为 Intel Visual Fortran Compiler 2013 或其更新的版本。虽然该程序是针对槽式聚光器编写而成的，但是读者也可以通过适当的修改来模拟其他类型聚光器的光学性能。

1.2 MCRT 程序 TOPS 的主要变量表及源程序

MCRT 程序 TOPS 源代码的主要变量参见表 1。

表 1 TOPS 的主要变量表

| 变量名 | 变量含义 |
|---------------|---------------------------------|
| Deltasun | 太阳视半径 |
| WT | 每根光线代表的辐射功率份额，初始发射的光线的功率份额为 1.0 |
| Power_per_ray | 每根光线代表的辐射功率值 |
| Nray | 参与追迹的光线总数 |
| DNI | 阳光直射辐照度 |
| Wm | 镜宽 |
| Lm | 镜长 |
| Fm | 镜的焦距 |
| Ref_M | 镜的反射率 |
| SigmaM | 镜面的形面误差的标准差 |
| Mpoint | 光线与镜面交点的坐标 |
| Incident | 入射光线单位向量 |
| Reflected | 反射光线单位向量 |
| La | 吸热管长度 |
| Ra | 吸热管外半径 |
| ABpoint | 光线与吸热管外壁交点的坐标 |
| ABpoint_Deg | 光线与吸热管外壁交点对应的圆周方向的角度 |
| Absor_AB | 吸热管外壁对阳光的吸收率 |
| ABHitorNot | 用于标记是否击中吸热管的变量 |
| StatAB_WT | 统计管外壁吸收的辐射功率份额的变量 |
| Nx | 管长度方向的网格线的数量 |
| Nc | 管圆周方向的网格线的数量 |

续表

| 变量名 | 变量含义 |
|---------|--------------------------|
| XlineAB | 存储管外壁 X 方向网格线的坐标值的数组 |
| YlineAB | 存储管外壁圆周方向网格线的 Y 坐标值的数组 |
| ZlineAB | 存储管外壁圆周方向网格线的 Z 坐标值的数组 |
| ClineAB | 存储管外壁圆周方向网格线的角度值的数组 |
| Pho_AB | 存储管外壁网格吸收辐射功率份额和热流密度值的数组 |

MCRT 程序 TOPS 主要包括 1 个主程序和 7 个子程序，其中主程序名为 TOPS；子程序 1 为 Define_AB_Grid，其作用为生成吸热管外壁上的矩形网格；子程序 2 为 Ray_Emission，其作用为发射光线；子程序 3 为 Reflection_on_M，其作用为计算光线在反射镜上的反射过程；子程序 4 为 Hit_AB，其作用为计算光线与吸热管外壁的相交作用；子程序 5 为 ABPoint_to_Deg，其作用为计算吸热管外壁上的点在圆周方向上对应的角度值；子程序 6 为 Stat_AB_Photon，其作用为统计吸热管外壁上每个网格所吸收的辐射功率份额；子程序 7 为 Stat_AB_Flux，其作用为计算吸热管外壁上每个网格吸收的热流密度值，并按一定的格式输出计算结果。具体的源程序如下：

```

!*****槽式聚光器光学性能的蒙特卡罗光线追迹模拟程序*****!
!*****Trough Optical Performance Simulation(TOPS)*****!
!定义用于生成[0,1)内均匀分布随机数的模块
Module Random_No_Mod
  Implicit None
  contains
  Function ran()
    implicit none
    integer, save:: flag=0
    double precision:: ran
    if(flag==0) then
      call random_seed()
      flag=1
    endif
    call random_number(ran)
  end Function ran
End Module Random_No_Mod
!*****!
Program TOPS          !主程序开始
  use Random_No_Mod  !生成[0,1)内均匀分布随机数的模块

```

```

    Implicit None
!声明常量
    real,parameter:: PI=3.14159265
!声明计算辅助参数
    real Time,Time_Begin,Time_End !计算计算时长
    integer(8) j !整型变量
!声明太阳参数
    real:: Deltasun !太阳视半径(阳光最大不平行半角)
    real(8):: WT !每根光线代表的功率份额:初始光线的功率份额为1.0
    real(8):: Power_per_ray !每根光线代表的辐射功率
    real(8):: Nray !参与追迹的光线总数
    real(8):: DNI !阳光直射辐照度
!声明聚光镜参数
    real:: Wm,Lm,Fm !镜宽、长、焦距
    real:: Ref_M,SigmaM !镜的反射率、形面误差的标准差
    real:: Mpoint(3,1),Point(3,1) !光线与镜面交点坐标
    real:: Incident(3,1),Reflected(3,1) !入、反射光线向量
!声明吸热管参数
    real:: La !管长
    real:: Ra !管外径
    real:: ABpoint(3,1) !光线击中管外壁的点
    real:: ABpoint_Deg !点对应的角度值,以管最低处为0°,逆时针旋转为正
    real:: Absor_AB !吸热管外壁的吸收率
    integer ABHitorNot !判断光线是否击中管的变量
    real(8)::StatAB_WT=0!统计管吸收的功率份额
!声明吸热管外壁的网络数组、热流密度统计数组
    integer:: Nx,Nc !管长度方向、圆周方向网格线数
    real,allocatable:: XlineAB(:),YlineAB(:),ZlineAB(:)
        !管网格线对应的X、Y、Z坐标
    real,allocatable:: ClineAB(:) !管周向网格线的角度
    real,allocatable:: Pho_AB(:, :) !统计管外壁网格吸收功率份额的数组

!太阳参数赋值
    Nray=1e10 !追迹的总光线数
    Deltasun=0.0075 !太阳视半径, rad
    DNI=1000 !直射辐照度 W/m²
!聚光镜参数赋值
    Lm=4.0 !镜长度, m

```

```

Wm=4.39823          !镜宽度, m
Fm=1.09957         !镜焦距, m
Ref_M=1            !镜反射率
SigmaM=0           !镜面形面误差的标准差, rad
!吸热管参数赋值
Ra=0.035           !管半径, m
La=4.0             !管长, m
Absor_AB=1         !管吸收率
Power_per_ray=dbl( (Lm*Wm*DNI) /Nray) !每根光线携带的辐射功率, W
Nx=321             !吸热管长度方向网格线数目
Nc=120             !吸热管圆周方向网格线数目
!创建可分配的管网格数组
allocate(XlineAB(Nx), YlineAB(Nc), ZlineAB(Nc), ClineAB(Nc))
!创建可分配的用于统计管外壁吸收的功率份额的数组
allocate(Pho_AB(Nx-1, Nc))
Pho_AB=0           !数组归零
!调用子程序Define_AB_Grid, 生成吸热管外壁的四边形网格
Call Define_AB_Grid(Nx, Nc, La, Ra, XlineAB, YlineAB, ZlineAB,
ClineAB)

!开始进行光线追迹
Call Cpu_Time(Time_Begin) !标记开始计算的时刻
print*, '程序正在运行, 请稍候...' !输出提示语句
Do j=1, Nray        !追迹第j条光线
  WT=1.0            !默认初始发射的光线携带的功率份额为1.0
  if(mod(j, 10**6)==0.or.j==1.or.j==Nray) then
    print '(A10, I12, A12)', '正在追迹第', j, '根光线'
  endif
!调用Ray_Emission子程序, 在XYZ系中计算不平行入射光线及其与镜面交点
call Ray_Emission(Incident, Wm, Lm, Fm, Mpoint, Deltasun)
!判断入射光线是否直接击中吸热管
Point=Mpoint       !将Mpoint的值赋值给临时变量Point
call Hit_AB(Incident, Point, Ra, ABpoint, ABHitorNot, La)
if(ABHitorNot==1) then !若入射光线直接击中吸热管
  call ABpoint_to_Deg(ABpoint, ABpoint_Deg) !计算点所在角度
  goto 0001         !接着, 去判断光线是否被吸收
endif
!若入射光线没有直接击中吸热管, 那么接着计算光线在镜上的反射过程

```

```

Mpoint=Mpoint           !接着使用预存的入射光线与镜面交点
if (ran()>Ref_M) then !判断光线是否被镜反射
    goto 0002           !若入射光线未被反射，则放弃之
endif
!若入射光线被反射，则继续计算光线在镜上的反射
    call Reflection_on_M(Incident,Reflected,Mpoint,Fm,SigmaM)
!接着，判断被镜反射的光线是否击中吸热管
    call Hit_AB(Reflected,Mpoint,Ra,ABpoint,ABHitorNot,La)
if (ABHitorNot==1) then !若反射光线击中吸热管
    call ABpoint_to_Deg(ABpoint,ABpoint_Deg)!计算点所在角度
    goto 0001           !去判断光线是否被吸收
else
    goto 0002           !若反射光线未击中吸热管，则放弃之
endif

!接着，统计吸热管外壁吸收的功率份额
0001   WT=WT*Absor_AB           !管外壁吸收的功率份额
       StatAB_WT=StatAB_WT+WT!管外壁吸收的总功率份额
       call Stat_AB_Photon(WT,Nx,Nc,XlineAB,ClineAB,ABpoint,
                           ABpoint_Deg,Pho_AB)           !统计网格吸收功率份额
       goto 0002           !继续追迹下一条光线
0002   continue               !继续追迹下一条光线
End Do                               !第j条光线追迹结束
Call Cpu_Time(Time_End)             !标记结束时刻的时间
Time=Time_End-Time_Begin            !计算光线追迹时长(秒)

!显示蒙特卡罗光线追迹计算结果
print*, '计算时长=', Time, '秒'
print*, 'DNI=', DNI
print*, '追迹的光线总数=', Nray
print*, '管吸收的光线数=', StatAB_WT
print*, '入射直射辐射的总功率=', Nray*Power_per_ray, 'W'
print*, '管外壁吸收的辐射功率=', StatAB_WT*Power_per_ray, 'W'
print*, '槽式聚光器的光学效率=', StatAB_WT/Nray*100, '%'

!生成输出结果所需文件
open(4, file='Power_and_Efficiency.dat', form='formatted')
write(4, '(A21,f20.3,A2)') '入射直射辐射的总功率=',
Nray*Power_per_ray, 'W'

```

```

write(4, '(A21,f20.3,A2)') '管外壁吸收的辐射功率=',
StatAB_WT*Power_per_ray, 'W'
write(4, '(A21,f20.3,A2)') '槽式聚光器的光学效率=',
StatAB_WT/Nray*100, '%'
close(4)
!调用子程序Stat_AB_Flux, 输出吸热管外壁吸收的辐射热流密度分布
call Stat_AB_Flux(Nx,Nc,La,Ra,XlineAB,YlineAB,ZlineAB,
ClineAB,Pho_AB, Power_per_ray,DNI)
pause !暂停
end program TOPS !主程序结束
!*****!
!*****吸热管外壁网格生成子程序*****!
Subroutine Define_AB_Grid(Nx,Nc,La,Ra,XlineAB,YlineAB,ZlineAB,
ClineAB)
Implicit None
integer Nx,Nc,i,j
real ClineAB(Nc),XlineAB(Nx),YlineAB(Nc),ZlineAB(Nc),La,Ra
real,parameter ::PI=3.14159265
!计算x方向网格线坐标
Do i=1,Nx
XlineAB(i)=(i-1)*La/(dble(Nx)-1.0)
enddo

Do j=1,Nc
ClineAB(j)=(j-1)*2*PI/dble(Nc)-PI!计算圆周方向网格线的角度
YlineAB(j)=Ra*sin(ClineAB(j)) !计算圆周方向网格线y坐标值
ZlineAB(j)=-Ra*cos(ClineAB(j)) !计算圆周方向网格线z坐标值
enddo
End Subroutine Define_AB_Grid
!*****!
!*****不平行光线发射子程序*****!
Subroutine Ray_Emission(Incident,Wm,Lm,Fm,Mpoint,Deltasun)
use Random_No_Mod !使用生成[0,1)内均匀分布随机数的模块
real,parameter ::PI=3.14159265
real Incident(3,1),Mpoint(3,1) !入射向量及其与镜交点
real Wm,Lm,Fm !镜宽、长、焦距

```

```

real Deltasun          !太阳视半径,rad
real Rhosun,Thetasun  !随机光线径向、切向偏角
!确定XYZ中光线与镜面交点的坐标Mpoint
Mpoint(1,1)=Lm*ran()
Mpoint(2,1)=-Wm/2.0+Wm*ran()
Mpoint(3,1)=Mpoint(2,1)**2.0/Fm/4.0-Fm
!计算随机入射光线的径向、切向偏角
Rhosun=asin(((sin(Deltasun))**2.0*ran())**0.5) !径向偏角
Thetasun=2.0*PI*ran() !切向偏角
!求得XYZ中的入射向量Incident
Incident(1,1)=Rhosun*cos(Thetasun)
Incident(2,1)=Rhosun*sin(Thetasun)
Incident(3,1)=-sqrt(1.0-Incident(1,1)**2.0-Incident(2,1)
**2.0)
end Subroutine Ray_Emission
!*****!

!*****计算光线在反射镜上的反射过程*****!
Subroutine Reflection on M(Incident,Reflected,Mpoint,Fm,SigmaM)
use Random_No_Mod !使用生成[0,1)内均匀分布随机数的模块
Implicit None
real,parameter ::PI=3.14159265
real Incident(3,1),Reflected(3,1) !入射、反射向量
real Mpoint(3,1) !入射光线与镜面交点
real Fm !镜焦距
real N(3,1) !Mpoint处的法向量
real M(3,3) !3*3矩阵
real Phi,Theta !理想法向量的径向偏角、切向偏角
real SigmaM !镜面型面误差的标准差,rad
real k !理想镜面型线在Mpoint处的斜率
real DotProduct !点积值
!计算局部坐标系X1Y1Z1中的理想法向量N
Phi=2*PI*ran() !切向偏角
Theta=sqrt(-2*SigmaM**2*log(1-ran())) !径向偏角
N(1,1)=Theta*cos(Phi) !N的X1分量
N(2,1)=Theta*sin(Phi) !N的Y1分量
N(3,1)=sqrt(1.0-N(1,1)*N(1,1)-N(2,1)*N(2,1)) !N的Z1分量
k=Mpoint(2,1)/2.0/Fm !求理想镜面型线在Mpoint处的斜率

```

```

k=atan(k)                !将k转换为Mpoint处型线的倾角
!将x1y1z1中的理想法向量N转换为XYZ中的实际法向量N
M=0                      !给坐标转换矩阵M赋值
M(1,1)=1
M(2,2)= cos(k)
M(3,2)= sin(k)
M(2,3)=-sin(k)
M(3,3)= cos(k)
N=MATMUL(M,N)          !求得XYZ中的实际法向量N
DotProduct=Incident(1,1)*N(1,1)+Incident(2,1)*N(2,1)+
Incident(3,1)*N(3,1)    !Incident与N的点积
Reflected=-2*DotProduct*N+incident !求得XYZ下反射的向量
end Subroutine Reflection_on_M
!*****!

!*****计算光线与吸热管外壁相交作用*****!
Subroutine Hit_AB(Reflected,Mpoint,Ra,ABpoint,ABHitorNot,La)
  Implicit None
  real Reflected(3,1),Mpoint(3,1),ABpoint(3,1)
  real Ra
  real(8) k,b            !光线一次方程参数
  real(8) Delt          !二次函数求根公式的Δ
  real(8) y1,y2,z1,z2,x,y,z !光线与管交点坐标值
  real La
  integer ABHitorNot
  !确定光线方程：因存在随机误差，可认为Reflected(2,1)不会为0.
  k=Reflected(3,1)/Reflected(2,1)
  b=Mpoint(3,1)-Reflected(3,1)/Reflected(2,1)*Mpoint(2,1)
  !求解直线与圆交点
  Delt=(2*k*b)**2-4*(1+k*k)*(b*b-dble(Ra)*dble(Ra)) !计算Δ值
  !若有2或1个交点，则计算之
  if(Delt>=0) then
    y1=(-2*k*b+sqrt(Delt))/2.0/(1+k**2)
    y2=(-2*k*b-sqrt(Delt))/2.0/(1+k**2)
    z1=k*y1+b
    z2=k*y2+b
    if(Reflected(3,1)<0) then
      if(z1>z2) then

```

```
        z=z1
        y=y1
    else
        z=z2
        y=y2
    endif
elseif (Reflected(3,1)>0) then
    if(z1<z2) then
        z=z1
        y=y1
    else
        z=z2
        y=y2
    endif
elseif(Reflected(3,1)==0) then
    if(Reflected(2,1)>0) then
        if(y1<y2) then
            y=y1
            z=z1
        else
            y=y2
            z=z2
        endif
    elseif(Reflected(2,1)<0) then
        if(y1>y2) then
            y=y1
            z=z1
        else
            y=y2
            z=z2
        endif
    else
        print*, 'Error in Hit_AB' !输出错误提示
        pause
    endif
endif
endif
```

!求得光线交点在xyz中的x坐标

```

x=Mpoint(1,1)+(y-Mpoint(2,1))*Reflected(1,1)/Reflected(
2,1)
if(x>=0.and.x<=La)then !说明交点位于吸热管范围内
    ABHitorNot=1 !将用于标记击中与否的变量赋值为1
else !说明交点位于吸热管范围外
    ABHitorNot=0 !将用于标记击中与否的变量赋值为0
endif
!求得光线与吸热管交点的坐标ABpoint
ABpoint(1,1)=x
ABpoint(2,1)=y
ABpoint(3,1)=z
else !说明光线未击中吸热管
    ABHitorNot=0 !将用于标记击中与否的变量赋值为0
endif
End Subroutine
!*****!
!*****计算吸热管外壁上的点在圆周方向对应的角度值*****!
Subroutine ABpoint_to_Deg(ABpoint,ABpoint_Deg)
    Implicit None
    real ABpoint(3,1) !光线击中管的点
    real ABpoint_Deg !光线击中管点对应的周向角度
    real Ra !吸热管外半径
    real,parameter ::PI=3.14159265
    !计算ABpoint对应的周向角度
    Ra=sqrt(ABpoint(2,1)**2.0+ABpoint(3,1)**2.0)!吸热管外半径
    if(ABpoint(2,1)>=0)then
        ABpoint_Deg=acos(-ABpoint(3,1)/Ra)
    elseif(ABpoint(2,1)<0)then
        ABpoint_Deg=-acos(-ABpoint(3,1)/Ra)
    endif
End Subroutine ABpoint_to_Deg
!*****!
!*****统计吸热管外壁上每个网格所吸收的辐射功率份额*****!
Subroutine Stat_AB_Photon(WT,Nx,Nc,XlineAB,ClineAB,ABpoint,
ABpoint deg,Pho AB)
    Implicit None

```

```

integer i,j
integer Nx,Nc
real(8) WT
real XlineAB(Nx),ClineAB(Nc),ABpoint(3,1),ABpoint_deg
real Pho_AB(Nx-1,Nc)
real,parameter ::PI=3.14159265
!判断交点是否在x方向上的某两根网格线间
Do i=1,Nx-1
  if (ABpoint(1,1)>=XlineAB(i).and.ABpoint(1,1)
    <XlineAB(i+1)) then
    Goto 1000
  endif
enddo
if (i==Nx) then
  Goto 3000
endif
1000 continue
!判断交点是否在圆周方向上的某两根网格线间
Do j=1,Nc-1
  if (ABpoint_deg>=ClineAB(j).and.ABpoint_deg
    <ClineAB(j+1)) then
    Goto 2000
  endif
enddo
if(ABpoint_deg>=ClineAB(Nc).and.ABpoint_deg<2*PI) then
  !在第Nc、1根网格线之间
  j=Nc
endif
2000 Pho_AB(i,j)=Pho_AB(i,j)+WT !相应(i,j)网格光线数+WT
3000 continue
end Subroutine Stat_AB_Photon
!*****!

!*****计算吸热管外壁上每个网格的热流密度,并按一定的格式输出计算结果*****!
Subroutine Stat_AB_Flux(Nx,Nc,La,Ra,XlineAB,YlineAB,ZlineAB,
ClineAB,Pho_AB, Power_per_ray,DNI)
  Implicit None
  real,parameter ::PI=3.14159265

```

```

integer i,j
integer Nx,Nc
real ClineAB(Nc),XlineAB(Nx),YlineAB(Nc),ZlineAB(Nc)
real Pho_AB(Nx-1,Nc),La,Ra
real(8) DNI !阳光直射辐照度,W·m^2
real(8) Power_per_ray
real(8) Dx,Drad,Dc,Ds
integer P1,P2,P3,P4 !每个网格的4个顶点的临时编号
real Theta
real Temp

!打开文件,用于输出吸热管外壁的局部聚光比LCR
open(1,file='Absorber_Local_Concentration_Ratio.csv',form='
formatted')
!打开文件,用于按照Tecplot 360软件格式输出吸热管外壁的热流密度
open(2,file='Absorber_Flux_Tecplot.dat',form='formatted')
!打开文件,用于按照ANSYS FLUENT软件中的Profile的格式要求,输出管外壁
每个网格在10^-6m的厚度内沿厚度方向均匀分布的热源值
open(3,file='Absorber_Source_Fluent_Profile.csv',form='form
atted')

!接着,计算网格的几何参数
Drad=(ClineAB(2)-ClineAB(1))/2.0 !周向相邻网格的弧度间隔
Dx=La/(dble(Nx-1))/2.0 !x方向的网格宽度
Dc=(ClineAB(2)-ClineAB(1))*Ra/2.0 !每个网格的圆弧半长度
Ds=Dx*Dc*4.0 !每个网格弧面的面积

!接着,输出吸热管外壁每个网格处的局部聚光比LCR
write(1,'(A20,A20,A10)') '网格中心X坐标/m','网格中心角度
/°','LCR,'
Do i=1,Nx-1
Do j=1,Nc
Pho_AB(i,j)=Pho_AB(i,j)*Power_per_ray/Ds
write(1,'(f14.8,A1)',advance='no') XlineAB(i)+Dx,', '
write(1,'(f14.8,A1)',advance='no')
(ClineAB(j)+Drad)*180.0/PI,', '
write(1,'(E24.12,A1)') Pho_AB(i,j)/DNI,', ' !输出LCR
enddo

```

```
endDo
```

!接着,按照ANSYS FLUENT软件中的Profile的格式要求,输出管外壁每个网格在 10^{-6} m的厚度内沿厚度方向均匀分布的热源值

```
write(3, '(A10)') '[Name],,,,'
write(3, '(A11)') 'Absorber,,,,'
write(3, '(A3)') ',,,,'
write(3, '(A9)') '[Data],,,,'
write(3, '(A20)') 'x,y,z,Source'
Do i=1,Nx-1
  Do j=1,Nc
    write(3, '(f14.6,A1)', advance='no') XlineAB(i)+Dx, ',,'
    if(j<Nc) then
      Temp=j+1
    else
      Temp=1
    endif
    write(3, '(f14.6,A1)', advance='no')
(-sin(ClineAB(j))-sin(ClineAB(Temp)))*Ra/2.0, ',,'
    write(3, '(f14.6,A1)', advance='no')
(cos(ClineAB(j))+cos(ClineAB(Temp)))*Ra/2.0, ',,'
    write(3, '(ES24.12)') Pho_AB(i,j)/1.0e-6 !输出热源值
  enddo
endDo
```

!接着,按照Tecplot 360软件格式输出吸热管外壁的热流密度

```
write(2,*) 'TITLE = "Absorber_Outer_Wall_Flux_Tecplot"'
write(2,*) 'VARIABLES = X, Y, Z, q/Wm^2'
write(2,*) 'ZONE T="Absorber Flux Density"'
write(2, '(A5,I8,2X,A5,I8,A30)') 'N=',Nx*Nc,',E=',(Nx-1)*Nc,',
ET=QUADRILATERAL, F=FEBLOCK'
write(2,*) 'VARLOCATION=( [1,2,3]=NODAL, [4]=CELLCENTERED) '
write(2,*) 'DT=(SINGLE SINGLE SINGLE SINGLE) '
!输出每个网格顶点的X、Y、Z坐标值以及热流密度值
Do i=1,Nc
  Do j=1,Nx
    write(2, '(f14.4)') XlineAB(j) !输出x坐标值
  enddo
```

```

enddo
Do i=1,Nc
  Do j=1,Nx
    write(2,'(f14.4)') YlineAB(i) !输出Y坐标值
  enddo
enddo
Do i=1,Nc
  Do j=1,Nx
    write(2,'(f14.4)') ZlineAB(i) !输出z坐标值
  enddo
enddo
Do i=1,Nc
  Do j=1,Nx-1
    write(2,'(E24.12)') Pho_AB(j,i) !输出热流密度
  enddo
enddo
!输出(Nx-1)*Nc个网格的4个顶点的序号
Do i=1,Nc
  Do j=1,Nx-1
    P1=j+(i-1)*Nx
    P2=j+(i-1)*Nx+Nx
    P3=j+(i-1)*Nx+Nx+1
    P4=j+(i-1)*Nx+1
    if(i==Nc) then
      P2=j
      P3=j+1
    endif
    write(2,'(4I8)') P1,P2,P3,P4 !输出网格顶点序号
  enddo
enddo
!关闭文件1-3
close(1)
close(2)
close(3)
end Subroutine Stat_AB_Flux
!*****!

```

1.3 MCRT 程序 TOPS 的模拟结果

TOPS 程序在完成模拟后，会生成四个存有结果的文件。具体而言：

文件 1, Power_and_Efficiency.dat 给出了入射到聚光器上的直射辐射的总功率、管外壁吸收的辐射功率以及槽式聚光器的光学效率。在本算例中，上述三个参数的值分别为 17592.92 W、17573.12 W 和 99.89%。

文件 2, Absorber_Flux_Tecplot.dat 是按照 Tecplot 360 软件的格式要求输出的管外壁热流密度(q_1)分布的数据。采用 Tecplot 360 软件，打开文件 2 后可获得如图 2 所示的热流密度分布图。由图 2 可见，吸热管外壁上的热流密度分布在圆周方向具有极大的非均匀性，而在轴向上则基本保持不变。

文件 3, Absorber_Local_Concentration_Ratio.csv 给出了管外壁每个矩形网格的局部聚光比值，同时还给出了每个矩形网格中心的 X 坐标值和在圆周方向上的角度值。

文件 4, Absorber_Source_Fluent_Profile.csv 按照 ANSYS FLUENT 软件中的 Profile 的格式要求，输出了管外壁每个网格在 10^{-6} m 的厚度内、沿厚度方向均匀分布的热源值。

为了说明 TOPS 对槽式聚光器光学性能模拟结果的准确性，图 3 对比了 TOPS 模拟获得的吸热管外壁典型的周向局部聚光比分布以及 Jeter^[1]采用数值积分方法获得的结果。从图 3 可以看出，TOPS 的模拟结果与 Jeter 的结果符合得很好，两条曲线几乎重叠，因而可认为本附录所介绍的 TOPS 程序是可靠的。




图 2 TOPS 模拟获得的吸热管外壁热流密度分布




图 3 TOPS 模拟获得的局部聚光比与 Jeter 的结果^[1]的对比

1.4 MCRT-FVM 耦合处理

在尽可能接近真实的非均匀能量分布条件下来研究太阳能吸热器中的光热转换过程，对于分析吸热器传热特性、发现系统缺陷和改进系统结构等都有十分重要的意义。

鉴于此，在采用 1.2 节所述的 MCRT 程序 TOPS，获得 1.3 节所述的吸热管外壁上的热流密度分布之后，需要采用 MCRT-FVM 耦合处理方法，将其准确无误地传递到用于模拟吸热管内的流动传热过程的有限容积法(FVM)模型中。对本算例而言，可采用如下步骤进行数据传递。

首先，在 1.2 节中，MCRT 程序 TOPS 在吸热管外壁上划分出了规则的矩形网格用于计算管外壁吸收的热流密度。而在 1.3 节中，TOPS 在完成模拟之后，已按照 ANSYS FLUENT 软件中的 Profile 的格式要求，输出了管外壁每个矩形网格在 10^{-6} m 的厚度内、沿厚度方向均匀分布的热源值，并存储在文件 4，即 Absorber_Source_Fluent_Profile.csv 中。

接着，采用 ANSYS FLUENT 软件建立吸热管的流动传热 FVM 模型。在建立 FVM 模型的过程中，需采用 ICEM CFD、GAMBIT 等商用软件在吸热管外壁上划分出与 TOPS 中一致的矩形网格。

最后，在 FVM 模型中将文件 4 以 Profile 的形式读入 ANSYS FLUENT，并将其作为能量方程的源项施加到吸热管外壁网格的 10^{-6} m 的薄层内，从而将吸热管外壁网格上的热流密度分布，由 MCRT 传递到 FVM。

基于上述方法，最终可实现 MCRT-FVM 耦合模型的两部分(MCRT 部分与 FVM 部分)网格之间的数据传递。

参 考 文 献

- [1] Jeter M S. Calculation of the concentrated flux density distribution in parabolic trough collectors by a semifinite formulation[J]. Solar Energy, 1986, 37(5): 335-345.